Elements of Statistical SPICE Models

Ning Lu, Josef Watts, and Scott K. Springer
IBM Semiconductor Research and Development Center, Systems and Technology Group,
Essex Junction, VT 05452 USA lun@us.ibm.com

ABSTRACT

We discuss several elements of statistical SPICE models and present our solutions. We present (i) a solution of automatically detecting Monte Carlo vs. skewed simulations, (ii) a method of modeling an arbitrarily given asymmetric or symmetric distribution, (iii) a hierarchical structure of the skewing parameters of many process/device statistical distributions for skewed simulations, (iv) a technique of combining chip-mean and across-chip variations for a single model parameter, (v) a method of correctly combining a chip-mean variation and an across-chip variation which is characterized as a percentage, and (vi) a method of enabling Monte Carlo simulations of across-chip variations at a skew corner of the chip-mean variations. These solutions establish a solid foundation for a good statistical SPICE model.

Keywords: SPICE modeling, statistical modeling, circuit simulation, asymmetric distribution.

1 INTRODUCTION

In VLSI circuit simulations, SPICE models play an important role. As semiconductor feature size decreases, the statisitical variations in circuit's characteristic, caused by statistical variations in semiconductor processes, become increasingly important. To achieve a robust circuit design and to improve circuit yield, Monte Carlo simulation capability has been added to SPICE simulators, and SPICE models have been expanded to include Monte Carlo statistical models and corner models [1–3]. In this paper, we discuss the building blocks needed for a set of high-quality statistical SPICE models, and present our solutions. These solutions can be applied to semiconductor FET compact models [2], to passive device models [4], and to SPICE models for enabling statistical modeling of parasitic resistance and capacitance elements (e.g., interconnect resistance and capacitance elements) in a SPICE netlist [5, 6].

2 AUTOMATIC DETECTION OF MONTE CARLO VS. SKEWED/CORNER SIMULATIONS

A good statistical SPICE model supports both Monte Carlo simulations and skewed/corner simulations in the same model set. To know whether a user will run Monte Carlo or skewed/corner simulations, often a model user is asked to use a toggle switch (`mc_sw`) to indicate his/her intentions.

```
.param mc_sw = 0    $ 0 or 1
+ rs_f = rs_n + rs_3sig * (mc_sw *  \\
agauss(0,1,3) + (1 – mc_sw) * cor_res/3)
```

It is not uncommon that the model's toggle switch indicates Mont Carlo simulations but the user actually runs skewed/corner simulation in his/her netlist. To avoid the use of a toggle switch like `mc_sw`, another approach was to use the following code (notice: no toggle switch),

```
.param cor_res = 0.0
+ rs_f = rs_n \\
    + rs_3sig * agauss(cor_res/3, 1, 3)
```

and, in a Model Reference Guide, ask users not to run Monte Carlo simultions with `cor_res != 0`. This approach does not actually prevent the misuse of the model. So a solution of automatically detecting Monte Carlo vs. skewed/corner simulations is a good improvement in a statistical model's usability. The follow code fragment sets the value of a global (on-off) switch `mc_sw` at run time based on whether it is actually a Monte Carlo or skewed/corner simulation:

```
.param mc1 = agauss(0, 1, 3)
+      mc2 = agauss(0, 1, 3)
+      mc3 = agauss(0, 1, 3)
+ is_mc='mc1 != 0 || mc2 !=0 || mc3 !=0'
...
+ p1_f = 'is_mc ? p1_mc : p1_skewing'
+ p2_f = 'is_mc ? p2_mc : p2_skewing'
```

where `agauss(0, 1, 3)` generates a random number from a Gaussian distribution during Monte Carlo runs but always returns a zero for a skewed/corner simulation in many SPICE simulators. In IBM PowerSpice simulator, the `agauss()` function varies with a simulator-level global skewing parameter "Nominal Random Number" in skewed/corner simulations. In such a case, the above `is_mc` line is modified to become

```
+ is_mc = 'mc1 != mc2 || mc2 != mc3'
```

The second auto-detection method is more general.

3 MODELING ASYMMETRIC DISTRIBUTIONS

In semiconductor processes and devices as well as in circuit's macro models, one often encounters asymmetric distributions. Examples include diffused or poly resistor's end resistance, contact resistance and back-end-of-line (BEOL) via resistance. Sometimes an asymmetric

distribution can be severely asymmetric. The above are distributions known at model construction time. There are also cases in which a statistical distribution could be either asymmetric or symmetric when the model is used. An example is the parasitic capacitance distributions generated by a parasitic extraction (PEX) tool. A SPICE model in a compact model set which supports such a PEX tool needs to handle both asymmetric and symmetric distributions. A good asymmetric distribution solution needs to handle a distribution from symmetric to moderately asymmetric to very asymmetric. When a skewing parameter is continuously varied from one end to the other, the resulting behavior in the asymmetric distribution should change monotonically and smoothly.

For an asymmetric distribution y, let n be its nominal value, b be its 3σ best-case value, and w be its 3σ worst-case value. The "3σ best-case value" can be either the 3σ maximum value or the 3σ minimum value of the distribution y. When y is moderately asymmetric, such as when $[\max(b, w) - n]/[n - \min(b, w)]$ is less than 5, one could find many nonlinear mapping relations to map y to a symmetric distribution x. For example, $y = x^2$, $y = x^3$, $y = e^x$, $y = xe^{\alpha x}$, etc. However, most of them can not handle very asymmetric distributions without becoming non-monotonic. To handle a very asymmetric distribution y, we consider a solution (mapping equation) of the form,

$$y(g; b, n, w) = n + C\left(\frac{1}{1+\alpha g} - 1\right), \tag{1}$$

where g is a symmetric distribution centered at $g = 0$. It is easy to see that the mapping between y and g is always monotonic. The singularity of function $1/(1 + \alpha g)$ at $\alpha g = -1$ gives the capability of the mapping relation (1) to handle a very asymmetric distribution. By construction this equation satisfies:

$$y(0; b, n, w) = n\,,$$

We further require it to meet:

$$y(1; b, n, w) = b\,, \tag{2a}$$

$$y(-1; b, n, w) = w\,. \tag{2b}$$

It is straightforward to determine the two coefficients in (1),

$$\alpha = \frac{w+b-2n}{w-b}\,, \tag{3a}$$

$$C = \frac{2(w-n)(n-b)}{w+b-2n}\,. \tag{3b}$$

In Eq. (1), g is `agauss(0, 1, 3)` (i.e., a Gaussian distribution with mean 0 and standard deviation 1/3) in Monte Carlo simulations, and takes value 0 at nominal and varies between -1 at the 3σ worst case and $+1$ at the 3σ best case in skewed simulations. On the first glance, the mapping relation (1) is not applicable for a symmetric distribution in which $w + b = 2n$ because the equation for C is singular in this case. However, a further examination shows that the degenerate factor $(w + b - 2n)$ in both the denominator and numerator of Eq. (1) can be cancelled out, leading to

$$asym(g; b, n, w) = n + \frac{2(w-n)(b-n)g}{w-b+(w+b-2n)g}\,. \tag{4}$$

Figures 1 [6] and 2 illustrate a particular example of this mapping. When a given distribution is symmetric, $w + b = 2n$, asymmetric mapping relation (4) becomes symmetric,

$$sym(g; b, n, w) = n + (b-n)g\,. \tag{5}$$

The compact expression in (4) is easy for implementation in a SPICE model and is efficient in circuit simulations.

Other types of mapping relations which can handle very asymmetric distributions can be constructed in a form similar to Eq. (1) by using functions with singularities. The following are some examples:

$$\begin{aligned}
&y_1(g; b, n, w) = n + C\left(\frac{1}{(1+\alpha g)^m} - 1\right), \quad m > 0\,,\\
&y_2(g; b, n, w) = n + C\ln(1+\alpha g)\,,\\
&y_3(g; b, n, w) = n + C[\tan(\tfrac{1}{4}\pi + \alpha g) - 1]\,,\\
&y_4(g; b, n, w) = n + C[\mathrm{ctg}(\tfrac{1}{4}\pi + \alpha g) - 1]\,,\\
&y_5(g; b, n, w) = n + C[\sec(\tfrac{1}{4}\pi + \alpha g) - \sqrt{2}]\,,\\
&y_6(g; b, n, w) = n + C[\csc(\tfrac{1}{4}\pi + \alpha g) - \sqrt{2}]\,,\\
&y_7(g; b, n, w) = n + C[\mathrm{cth}(1+\alpha g) - \mathrm{cth}(1)]\,,\\
&y_8(g; b, n, w) = n + C[\mathrm{csch}(1+\alpha g) - \mathrm{csch}(1)]\,.
\end{aligned} \tag{6}$$

Two coefficients C and α are determined by two nonlinear equations in (2). Figure 3 plots several mapping relations in Eqs. (6) for the same b, n, and w values as in Fig. 2.

4 HIERARCHICAL STRUCTURE OF THE SKEWING PARAMETERS OF MANY PROCESS DISTRIBUTIONS

In skewed/corner simulations, for ease of use, it is preferred to limit the number of skewing parameters even though the number of underlying statistical process parameters is very large. For example in many circuits it may be sufficient to simply skew resistors between high and low resistance. However in some cases it may be important to skew one type of resistor high and another type low. More detailed control is also needed to support an optimal corner [7] and/or to support an efficient method of corner model generation. So we desire one independent skewing parameter for each independent Monte Carlo distribution. A solution for these two opposing requirements is to establish a hierarchical structure of the skewing parameters of many process distributions. By default, a higher-level skewing parameter controls many lower-level skewing parameters. When specified, however, a lower-level skewing

NSTI-Nanotech 2009, www.nsti.org, ISBN 978-1-4398-1784-1 Vol. 3, 2009

parameter takes precedence over its higher-level skewing parameter. The following code fragment is part of a hierarchical structure of skewing parameters. Here it is for a family of unsilicided resistors (using `cor_res`). Family members include N+ diffusion resistor and P+ poly resistor, etc. Each of resistor has a sheet resistance distribution (using `ndrs_sigma`, `ppcrs_sigma`), an end resistance distribution (using `ndre_sigma`, `ppcre_sigma`), etc.

```
.param cor_res = 0.0   $ Typical range:
+ ndrs_sigma  = cor_res   $ [-3, +3]
+ ndre_sigma  = cor_res
+ ppcrs_sigma = cor_res
+ ppcre_sigma = cor_res
...
+ ndrs_mc = agauss(0, 1, 3)
+ ndrs_f = 'ndrs_n + ndrs_3sig \\
  *(mc_sw ? ndrs_mc : ndrs_sigma / 3)'
+ ndre_mc = agauss(0, 1, 3)
+ ndre_1bm1w = 'mc_sw ? ndre_mc \\
                     : ndre_sigma / 3'
+ ndre_f = asym(ndre_1bm1w, \\
          ndre_min, ndre_n, ndre_max)
...
+ ppcrs_mc = agauss(0, 1, 3)
+ ppcrs_f = 'ppcrs_n + ppcrs_3sig \\
   *(mc_sw ? ppcrs_mc : ppcrs_sigma / 3)'
+ ppcre_mc = agauss(0, 1, 3)
+ ppcre_1bm1w = 'mc_sw ? ppcre_mc \\
                    : ppcre_sigma / 3'
+ ppcre_f = asym(ppcre_1bm1w, \\
         ppcre_min, ppcre_n, ppcre_max)
```

For a SPICE simulator (e.g., IBM PowerSpice) which does not allow overriding a variable (on the right side of an equal sign), the following code segment illustrates a second approach:

```
.param cor_res = 0.0
+ ndrs_sigma = 1e35 $an impossible value
...
+ ndrs_sigma_f = ndrs_sigma == 1e35 ? \\
                  cor_res : ndrs_sigma
+ ndrs_f = 'ndrs_n + ndrs_3sig \\
  *(mc_sw ? ndrs_mc : ndrs_sigma_f / 3)'
```

With the establishment of a hierarchical structure of the skewing parameters with an independent skewing parameter for each statistical model parameter in a compact model, designer A can easily explore the overall effect of variations of resistors' resistance on his circuit by varying the higher-level skewing parameters `cor_res`, etc. Such a structure also allows designer B or model developers first to find the sensitivity of a performance target on each statistical model parameter, and then to set each of the lower-level skewing parameters `ndrs_sigma`, `ndre_sigma`, `ppcrs_sigma`, and `ppcre_sigma`, etc. to different values (either to arrive at an optimal 3σ corner for a single performance target or to find a common 3σ corner for multiple performance targets).

5 MODELING THE COMBINED EFFECTS OF CHIP-MEAN AND ACROSS-CHIP VARIATIONS

Combining chip-mean and across-chip variations for a single model parameter. Both chip-mean (CM) and across-chip variations (ACV) of a model parameter p contribute to its total variations. Examples include FET channel length and width, Vth0, etc. In a statistical model, this is enabled by

$$p_i = n + \sigma_{cm} G + \sigma_{acv} g_i \,, \quad i = 1, 2, 3, \cdots, \tag{7}$$

when the variable p (say, Vth0) is treated as completely uncorrelated between different instances of the same type device. Here each of G and g_i is an independent normal distribution of mean 0 and standard deviation 1. The index i ranges over all instances of the parameter. Alternatively, when the parameter p (say, channel length and width) is treated as completely correlated between device instances, it is enabled by

$$p_i = n + \sigma_{cm} G + \sigma_{acv} g_0 \,, \quad i = 1, 2, 3, \cdots. \tag{8}$$

The total variation of p is $\sigma_p = \sqrt{\sigma_{cm}^2 + \sigma_{acv}^2}$ in both cases. For some model parameters (e.g., channel length and width) or circuits (e.g., ring oscillator) representation in a macro model), one can model them as random-correlated when two instances are located nearby but model them as random-uncorrelated when the two instances are located far apart. In between, one can model the degree of correlation decreases with increasing separation [8].

When an across-chip variation is characterized as a percentage of the mean value. Sometimes the amount of across-chip variation of a model parameter (e.g., the mobility in a FET model) or the mismatch amount of a device characteristic is characterized as a percentage of the mean value of the parameter. This is most often modeled as the product of the chip-mean part and a mismatch multiplier,

$$p_i = (n + \sigma_{cm} G)(1 + \sigma_{mm} g_i) \,, \quad i = 1, 2, 3, \cdots. \tag{9}$$

This modeling method, however, introduces an un-expected coupling ($\sigma_{cm} G \sigma_{mm} g_i$ term) between systematic (i.e., chip mean) variation $\sigma_{cm} G$ and random variation $\sigma_{mm} g_i$. The correct modeling approach should be [9]

$$p_i = n + \sigma_{cm} G + n \sigma_{mm} g_i \,, \quad i = 1, 2, 3, \cdots. \tag{10}$$

Enabling Monte Carlo simulations of across-chip variations at a chip-mean skew corner. Circuit designers often need to ensure the functionality of their circuits at the worst chip-mean corner and worst case across chip variation. This can be achieved by varying both chip mean and across variation in Monte Carlo. However this is a very unlikely case. A more efficient method is to combining Monte Carlo variation of across chip effects at the worst case chip mean corner,

$$p_i = n + \sigma_{cm} s + \sigma_{acv} g_i, \; i = 1, 2, 3, \ldots, \qquad (11)$$

where s is a skewing parameter. This capability in a SPICE model enables circuit designers to perform a smaller number of Monte Carlo runs at a chip-mean corner to obtain a larger sample of ACV.

6 SUMMARY

We have presented our modeling methodology and solutions on several elements of statistical SPICE models. Each of them addresses a particular issue in statistical SPICE modeling. When combined together, they enable a set of good statistical SPICE models. Many of these elements have been implemented in Common Platform's compact models and in Advanced SOI Technology Alliance's compact models.

ACKNOWLEDGMENT

The authors would like to thank IBM management for suppport.

REFERENCES

[1] C. C. McAndrew and P. G. Drennan, *Proc. NSTI Nanotech, Workshop on Compact Modeling*, pp. 715–718, 2002. C. C. McAndrew and P. G. Drennan, *Proc. NSTI Nanotech, Workshop on Compact Modeling*, pp. 698–702, 2006.

[2] J. Watts, N. Lu, C. Bittner, S. Grundon, and J. Oppold, *Proc. NSTI Nanotech, Workshop on Compact Modeling*, Vol. 3, pp. 87–92, 2005.

[3] S. K. Springer, S. Lee, N. Lu, E. J. Nowak, J.-O. Plouchart, J. S. Watts, R. Q. Williams, and N. Zamdmer, IEEE Trans. Electron Devices, **53**, pp. 2168–2178, 2006.

[4] C. C. McAndrew, *Proc. NSTI Nanotech, Workshop on Compact Modeling*, pp. 86–89, 2004.

[5] N. Lu, *Proc. IEEE Custom Integrated Circuits Conf.*, pp. 853–856, 2006.

[6] N. Lu and J. H. McCullen, *Proc. 8th International Symposium on Quality Electronic Design*, pp. 743–748, 2007.

[7] N. Lu, *Proc. NSTI Nanotech, Workshop on Compact Modeling*, Vol. 3, pp. 566–569, 2007.

[8] N. Lu, *Proc. NSTI Nanotech, Workshop on Compact Modeling*, Vol. 3, pp. 818–821, 2008.

[9] N. Lu, *Proc. 9th International Symposium on Quality Electronic Design*, pp. 543–548, 2008.



Fig. 1. Histogram plot of a Monte Carlo simulation result on a very asymmetric distribution of nominal value 1.85, 3σ minimum value 0.5, and 3σ maximum value 9.



Fig. 2. Mapping relation (4) from g to *asym* for the same asymmetric distribution as in Fig. 1.



Fig. 3. Mapping relations (6) from g to y for the same 3σ min., nominal, and 3σ max. values as in Figs. 1 and 2. At about g = –0.5, from top to bottom, six curves are y_7, y_5, y_1 (with m = 2), y_8, y_3, and y_2.